
Abstract. The Lloyd-Topor transformation is a classical transformation that
translates extended logic programs with logical connectives and explicit quantifiers
into normal logic programs. In this paper we show that this translation can be
achieved in a natural way by specialising a meta-interpreter for extended logic
programs. For this we use the logen partial evaluation system, extended to handle
coroutining.

1 Vanilla Self-interpreter

An interesting application for partial evaluation [7] is the specialisation of interpreters.
The object program to interpret is typically static and known at specialisation time,
while the runtime goal remains dynamic. Partial evaluation can remove the overhead of
interpretation, performing all the interpretation tasks at specialisation time and leaving
behind a much more efficient “compiled” program. A classic interpreter is the vanilla
self-interpreter (see, e.g., [1, 4]). Listing 1.1 contains a variation of the vanilla interpreter.
The predicate solve literal/1 looks up the clause definition in the clause database
and calls solve/1 its body. Clauses are represented by a head and a conjunction of
body literals. An auxiliary module (prolog reader) is used to load clauses from a file,
it defines load file/1 and get clause/2. Listing 1.1 has the entry point solve file/2
which takes as arguments the name of the file containing the clauses and an entry goal.

Listing 1.1. A Vanilla self-interpreter for Prolog
:- use_module(prolog_reader ).

solve(true).

solve(’,’(A,T)) :- solve(A), solve(T).

solve(A) :- nonvar(A), A\= ’,’(_,_), solve_literal(A).

solve_literal(A) :- prolog_reader:get_clause(A,B), solve(B).

solve_file(File ,A) :- prolog_reader:load_file(File),solve_literal(A).

While it may look like a simple program it still presents enough challenges for partial
evaluation and has attracted considerable attention (see, e.g., [8, 14, 17]). In recent work
[9] (see also [18]) we have shown how one can effectively specialise this interpreter using
an offline specialiser (such as logen [10]), achieving so-called Jones optimality [6, 13]. A
key ingredient for success was the careful annotation of the interpreter, using the nonvar
annotation which indicates that the top-level function symbol of some argument is known.

Indeed, the argument to solve literal/1 cannot be marked as static (as it can contain
variables), and marking it as dynamic would mean that no useful specialisation could
be achieved. In the rest of this paper we build upon this to reconstruct the classical
Lloyd-Topor transformation by specialising an interpreter derived from Listing 1.1.

2 The Lloyd Topor Transformation

Lloyd and Topor [11, 12] introduced extended programs and goals for logic programming.
The extended programs can contain clauses that have an arbitrary first-order formula
in their body. The only requirement for executing the transformed programs is a sound
form of the negation as failure rule. In [11, 12] an extended program P is transformed
into a normal program P ′, called the normal form of P , using a set of transformation
rules. The rules (a) . . . (j) are applied until no more transformations can be applied.
[11, 12] proves this process terminates and always gives a normal program.

(a) Replace A← α ∧ ¬(V ∧W ) ∧ β
by A← α ∧ ¬V ∧ β and A← α ∧ ¬W ∧ β

(b) Replace A← α ∧ ∀x1 . . . xnW ∧ β by A← α ∧ ¬∃x1 . . . xn¬W ∧ β

(c) Replace A← α ∧ ¬∀x1 . . . xnW ∧ β by A← α ∧ ∃x1 . . . xn¬W ∧ β

(d) Replace A← α ∧ V ←W ∧ β
by A← α ∧ V ∧ β and A← α ∧ ¬W ∧ β

(e) Replace A← α ∧ ¬(V ←W ) ∧ β by A← α ∧W ∧ ¬V ∧ β

(f) Replace A← α ∧ (V ∨W ) ∧ β by A← α ∧ V ∧ β and A← α ∧W ∧ β

(g) Replace A← α ∧ ¬(V ∨W ) ∧ β by A← α ∧ ¬V ∧ ¬W ∧ β

(h) Replace A← α ∧ ¬¬W ∧ β by A← α ∧W ∧ β

(i) Replace A← α ∧ ∃x1 . . . xnW ∧ β by A← α ∧W ∧ β

(j) Replace A← α ∧ ¬∃x1 . . . xnW ∧ β
by A← α ∧ ¬p(y1, . . . , yk ∧ β and p(y1, . . . , yk)← ∃x1 . . . xnW
where y1, . . . , yk are the free variables in ∃x1 . . . xnW and p is a new predicate not
already appearing in the program.

For example, take the definition of subset expressed as an extended program [11, 12]:

x ⊆ y ← ∀u(u ∈ y ← u ∈ x)

The definition is written in a clear mathematical way and has been expressed in
a form similar to the specification of the problem. However this specification cannot be
executed directly in Prolog as it contains ∀ and← in its body. Transforming the extended
program using the transformation rules produces the following normal program:

x ⊆ y ← ¬p(x, y)

p(x, y)← ¬(u ∈ y) ∧ u ∈ x

This program requires a sound implementation of negation as failure, such as the
one provided in [5]. As most Prolog systems do not provide such a negation, we can
implement a sound selection rule ourselves by delaying negative literals until they have
become ground using when/2. The so-obtained normal program, rewritten in standard
Prolog syntax is given in Listing 1.2 (also containing sample executions). The ∈ operator
has been replaced by calls to mem/2. Below we show how this transformed program can
be obtained automatically by specialising an interpreter for extended programs.

Listing 1.2. Prolog version of normal subset program using coroutining
subset(X,Y) :- when(ground ([Y,X]),\+p(Y,X)).

p(X,Y) :- when(ground ([A,X]),\+mem(A,X)), mem(A,Y).

mem(A, [A|_]).

mem(A, [_|B]) :- mem(A, B).

| ?- subset ([a,b,c], S), S = [d,e,f,a,b,c].

S = [d,e,f,a,b,c] ?

yes

3 An Interpreter for Extended Programs

We now extend our vanilla interpreter to handle exstended programs as well as a more
natural form of input programs. Note that this interpreter is not based on the Lloyd-
Topor transformation. First, new operators are defined for implication and negation in
the interpreter. The functions forall/2 and exists/2 are reserved for ∀ and ∃.
:- op(950,yfx ,’=>’).% implies right

:- op(950,yfx ,’<=’). % implies left

:- op(850,yfx ,’or’). % or

:- op(800,yfx ,’&’). % and

:- op(750,fy,’~’). % not

Two new predicates are created for handling body literals, a positive (solve literal/1)
and a negative one (not solve literal/1; introduced to overcome the problems of Pro-
log’s unsound negation). The definition for solve literal/1 contains the basic clauses
for dealing with true and false, if a not operator is encountered control is passed to
not solve literal/1. The & operator performs a conjunction of the two arguments and
the or operator performs a disjunction.

solve_literal(true).

solve_literal(false) :- fail.

solve_literal(’~’(L)) :- not_solve_literal(L).

solve_literal(’&’(A,B)) :- solve_literal(A), solve_literal(B).

solve_literal(or(A,_)) :- solve_literal(A).

solve_literal(or(_,B)) :- solve_literal(B).

solve_literal(A) :- is_user_pred(A),solve_atom(A).

solve_literal(A) :- is_built_in(A),call(A).

solve_atom(A) :- my_clause(A,B), solve(B).

The clauses for not solve literal/1 are similar, but define the negated counterparts
of solve literal/1. Notice the handling of & and or, DeMorgan’s laws ( ¬(A ∨ B) ≡
(¬A ∧ ¬B) and ¬(A ∧ B) ≡ (¬A ∨ ¬B)) are applied and solve literal/1 is called.
Coroutining is used to delay the negation using when/2.

not_solve_literal(true) :- solve_literal(false ).

not_solve_literal(false) :- solve_literal(true).

not_solve_literal(’~’(L)) :- solve_literal(L).

not_solve_literal(or(A,B)) :- solve_literal(’&’(’~’(A), ’~’(B))).

not_solve_literal(’&’(A,B)) :- solve_literal(or(’~’(A), ’~’(B))).

not_solve_literal(A) :- is_user_pred(A), not_solve_atom(A).

not_solve_literal(A) :- is_built_in(A),

term_variables(A,Vars), when(ground(Vars), \+( call(A))).

not_solve_atom(A) :-

term_variables(A,Vars), when(ground(Vars), \+( solve_atom(A))).

Implication is handled by transforming it into a disjunction (using A→ B ≡ (¬A∨B)
and A← B ≡ (A ∨ ¬B)) , which will in turn be handled by the previous clauses.

solve_literal(’=>’(A,B)) :- solve_literal(or(’~’(A),B)).

solve_literal(’<=’(A,B)) :- solve_literal(or(A,’~’(B))).

not_solve_literal(’=>’(A,B)) :- solve_literal(’~’(or(’~’(A),B))).

not_solve_literal(’<=’(A,B)) :- solve_literal(’~’(or(A,’~’(B)))).

The exists(X,A) operator is implemented by making a copy of the atom, A, and
renaming all occurrences of X. This is done to avoid name clashes and solve literal/1
is called on the copy. The forall(X,A) is transformed into a exists(X,~A) and the neg-
ative version of forall(X,A) is transformed into a positive exists using standard logic
laws. The negative version of exists/2 must be handled directly, a when/2 declaration
is added to ensure the negation is only selected when the arguments are instantiated.

solve_literal(exists(X,A)) :- rename(X,A,CopyA),solve_literal(CopyA).

solve_literal(forall(X,A)) :- not_solve_literal(exists(X,’~’(A))).

not_solve_literal(forall(X,A)) :- solve_literal(exists(X,’~’(A))).

not_solve_literal(exists(X,A)) :-

force_not_solve_literal(exists(X,A)).

force_not_solve_literal(G) :- get_free_variables(G,[],[], Vars),

when(ground(Vars), \+( solve_literal(G))).

4 Specialising the Interpreter

In order to apply the logen [10] offline specaliser we first had to extend it to handle
the when declarations found in the above interpreter. Basically, we allow a when to be
marked as static, i.e., the when will be executed at specialisation time and the guard
must be satisfied at some point during the specialisation, or as dynamic in wich case
the body of the declaration is specialised and then wrapped inside a when declaration.

Listing 1.3. Subset extended program in interpreter form
subset(A,B) :- forall(C,member(C,B)<=member(C,A)).

Specialising the interpreter for the subset extended program in Listing 1.3 produces
the residual program Listing 1.4. The specialised program is almost identical to the
program in Listing 1.2. All negations have been enclosed by when/2, this will ensure the
negation will only be performed when the goal is properly ground. This transformation

was performed by specialising an interpreter. Importantly the interpreter was not simply
an encoding of the transformation rules from [12] but an intuitive interpreter for Prolog
handling the extended program syntax.

Listing 1.4. Specialising the interpreter for the subset extended program (Listing 1.3)
solve_atom(subset(A,B)) :- solve_atom__0(A, B).

solve_atom__0(A,B) :- when(ground ([A,B]),\+ solve_literal__1(A,B,_)).

solve_literal__1(A,B,_) :-

when(ground ([B,C]), \+ member(C,B)), member(C,A).

Experimental results for the subset example can be found in Table 1, run on a 2.4
GHz Pentium 4 with 512 Mb memory, Gentoo 2.6 Linux and running SICStus 3.11.1.

Original Specialised Relative
Benchmark Program Size Program Size Program Size

Lloyd Topor 19877 bytes 1685 bytes 0.08

Specialisation Original Specialised Speedup
Benchmark Iterations Time Runtime Runtime

Lloyd Topor 100000 60ms 8580ms 2370ms 3.57

Table 1. Benchmark figures for the Lloyd Topor interpreter

As the logen system uses the cogen approach [10], we were able to produce a compiler
which can be used to compile extended logic programs into ordinary Prolog programs
from the command-line. For this we had to add an ability to logen to feed command-line
arguments into the specialization goal. In our case, the name of the file containing the
extended Prolog program is passed as a static argument to the solve file/2 predicate
(Listing 1.1). We have also extended the interpreter to handle command-line flags which
allow to turn co-routining and debugging on and off; if co-routining is turned off we would
get Listing 1.4 without when declarations. These programs cannot be (safely) run on most
Prologs, but can be easier to read and resemble more directly the result of the Lloyd-Topor
transformation. The debugging feature is useful as it presents debugging information in
terms of the original extended program, and not in terms of the transformed program.
More details, as well as more experimental results will be presented in the full paper.

We can now relate the output of the specialisation process more formally to the result
of the Lloyd-Topor transformation. Let E be an extended program. By E ↓p/n we denote
the subset of the clauses of E reachable in the predicate dependency graph from the
predicate p/n.

Proposition 1. Specialisation of the annotated interpreter for extended programs ter-
minates for any extended object program E and entry predicate p/n. After removing the
when declarations, the residual program is identical (up to predicate and variable renam-
ing) to the result obtained by the Lloyd-Topor transformation on E ↓p/n.

The proof proceeds by linking the transformation rules (a) – (i) to unfolding steps and
the transformation rule (j) to a a memoisations step of the specialisation process.

5 Conclusion

We have shown how one can reconstruct the Lloyd-Topor transformation in a systematic
way by specialising an interpreter for extended logic programs which uses coroutining.

In addition to providing an insight about this classic transformation, this provided an
interesting test case for partial evaluation; stepping up from vanilla on to more challenging
and practically useful interpreters. We have also obtained a command-line compiler for
extended programs. An interesting direction for future work would be to build upon our
work and develop an interpreter for ever more powerful languages, e.g., constructing a
compiler for Gödel [2, 5], Curry [3] or Escher. It is also interesting to see whether other
transformations (e.g., [15, 16]) can be obtained in a similar fashion by partial evaluation.

16. S. Muñoz Hernándex, J. Mariño, and J. J. Moreno-Navarro. Constructive intensional nega-
tion. In Y. Kameyama and P. J. Stuckey, editors, Proceedings FLOPS’04, LNCS 2998.
Springer-Verlag, April 2004.

17. W. Vanhoof and B. Martens. To parse or not to parse. In N. Fuchs, editor, Proceedings
LOPSTR’97, LNCS 1463, pages 322–342, Leuven, Belgium, July 1997.

18. Q. Wang, G. Gupta, and M. Leuschel. Towards provably correct code generation via horn
logical continuation semantics. In M. V. Hermenegildo and D. Cabeza, editors, Proceedings
PADL’05, LNCS 3350, pages 98–112. Springer, 2005.

6


